Étude d'une suite définie par récurrence

On considère la suite (u_n) définie sur $\mathbf{N}$ par $u_0 = 2$ et

$$\forall n \in \mathbf{N}, \quad u_{n+1} = \frac{2}{3}\left(u_n + \frac{1}{u_n^2}\right)$$

Étudier le comportement de la suite (u_n).

Commençons par définir la fonction qui se *cache* derrière cette suite, $f : x \mapsto \frac{2}{3}\left(x + \frac{1}{x^2}\right)$.

```
> f(x):=2/3*(x+1/x^2);
```

```
f(x):=2/3*(x+1/x^2);
```

```
> assume(x>0);
```

$$[x > 0]$$

L'intervalle $]0, +\infty[$ est *stable* par f, *i.e.* si $x \in]0, +\infty[$ alors $f(x)$ est défini et $f(x) \in]0, +\infty[$. Ceci permet de justifier l'*existence* de la suite u :

```
> u[n]:=f(u[n-1]);
```

$$u[n]:=f(u[n-1]);$$

```
> u[0]:2;
```

$$2$$

Calculons les premiers termes :

```
> valeurs:makelist(u[i],i,0,5);
```

$$\left[2, \frac{3}{2}, \frac{35}{27}, \frac{125116}{99225}, \frac{2935497269576521}{2329904227757400}, \frac{3794338057878074966090774550686696421419046876 1}{3011568112298068778040219113051418195557582010 0}\right]$$

Nous obtenons des rationnels, passons aux *flottants* :

```
> float(valeurs);
```

$$[2.0, 1.5, 1.296296296296296, 1.260932224741749, 1.259921860565926, 1.259921049895395]$$

Cela *semble* converger. Recherchons l'éventuelle limite de la suite, un *point fixe* de f.

```
> ptfixes:solve(f(x)=x);
```

$$\left[x=\frac{2^{\frac{1}{3}}\sqrt{3}\,i-2^{\frac{1}{3}}}{2},x=-\frac{2^{\frac{1}{3}}\sqrt{3}\,i+2^{\frac{1}{3}}}{2},x=2^{\frac{1}{3}}\right]$$

Il y a un seul point fixe réel, le troisième.

```
> float(ptfixes[3]);
```

$$x=1.259921049894873$$

u_5 est bien proche de ce point fixe ($\sqrt[3]{2}$)... tout en étant supérieur.

Regardons de plus près la fonction f, en particulier le signe de sa dérivée lorsque $x \geqslant \sqrt[3]{2}$, c'est à dire $x^3 \geqslant 2$.

```
> assume(x^3-2>=0);
```

$$\left[x^3 \geqslant 2\right]$$

```
> sign(diff(f(x),x));
```

positif *ou* nul

La dérivée de f est donc positive sur $I = [\sqrt[3]{2}, +\infty[$, f est croissante sur cet intervalle. Compte tenu que la borne inférieure de I est point fixe et qu'il est non borné à droite, il est stable par f. Autrement dit tout les termes de la suite (u_n) sont dans I dans la mesure ou le premier d'entre eux y est (récurrence). D'où :

$$\forall n \in \mathbf{N}, \quad u_n \geqslant \sqrt[3]{2}$$

Déterminons le signe de $f(x) - x$, toujours pour $x \geqslant \sqrt[3]{2}$:

```
> sign(f(x)-x);
```

négatif *ou* nul

D'où : $\forall n \in \mathbf{N}$, $u_{n+1} - u_n \leqslant 0$, la suite (u_n) est donc décroissante. Nous pouvons conclure : (u_n) est décroissante et minorée, elle est convergente (théorème), sa limite est la seule limite possible : $\sqrt[3]{2}$.



Sur cette figure, nous retrouvons l'illustration des propriétés misent en avant pour justifier la convergence de la suite (u_n).